COMPOSITIONS

(des verbes, adverbes, conjonctions et noms) *Robert Coquidé (15/05/2018)*

En mathématiques, on utilise des données généralement numériques souvent structurées en vecteurs, matrices, tenseurs …. Ces structures des données ont été très enrichies (voir chapitre "Structures des données en J").

En J on utilise des données (**NOMS**) utilisées comme arguments et résultats des fonctions (**VERBES**). Les opérateurs se nomment des **ADVERBES** ou des **CONJONCTIONS**. Si un nom, un verbe, un adverbe ou une conjonction est affecté à un identificateur (**ID**), il se nomme : **PRO-NOM**, **PRO-VERBE**, **PRO-ADVERBE** ou **PRO-CONJONCTION**.

Ex: **+** est un **verbe** (inclus dans le "noyau" de J).

(+/ .*) est un **verbe composite** (créé par le programmeur en utilisant les règles de composition): c'est le produit matriciel cher aux mathématiciens.

Pm =. +/ .* est un **pro-verbe** (utilisable comme le verbe précédent en faisant appel à son identificateur **Pm**).

Les mathématiciens connaissent le "produit de composition" de 2 fonctions d'une variable:

$$H = F \circ G \quad \Leftrightarrow \quad H(x) = F(G(x))$$

Ils connaissent également la "somme" et la "différence" de 2 fonctions:

$$H = F + G \quad \Leftrightarrow \quad H(x) = F(x) + G(x)$$

$$H = F - G \quad \Leftrightarrow \quad H(x) = F(x) - G(x)$$

Ainsi que la "réciproque" de certaines fonctions:

$$G = F^{-1} \quad \Leftrightarrow \quad G(F(x)) = x \text{ et } F(G(x)) = x$$

Et la dérivée d'une fonction:

$G = F'$ $\quad \Leftrightarrow \quad$ G(x) = pente de la tg à la courbe d'équation $Y=F(x)$

de même la dérivée d'ordre n:

$G=F^{(n)}$ définie par récurrence $G=H'$ avec $H=F^{(n-1)}$

Ils utilisent la notion "d'**opérateurs**" qui agissent dans un ensemble de fonctions:

$$G = OP\,(F) \qquad \text{et} \qquad H = F_1\ OP\ F_2$$

En **"J"** tous ces concepts **mathématiques** ont été considérablement généralisés, développés et étendus**:**

- Beaucoup s'appliquent aux données alphanumériques.
- De nombreux opérateurs s'appliquent aux verbes (ou **pro-verbes**) monadiques (fonctions d'une variable) mais également aux verbes (ou **pro-verbes**) dyadiques (par exemple aux lois de composition "internes" ou "externes").
- Un opérateur qui n'utilise qu'un seul argument est un **adverbe** (ou **pro-adverbe**).
- Un opérateur qui utilise 2 arguments est une **conjonction** (ou **pro-conjonction**).
- L'exécution d'un adverbe (ou pro-adverbe), d'une conjonction (ou pro-conjonction), génère un verbe (ou pro-verbe), ayant, sauf exceptions, 2 significations (l'une monadique, l'autre dyadique); ils peuvent utiliser comme arguments non seulement des verbes (ou pro-verbes) mais aussi des noms (ou pro-noms).
- L'expression mathématique F + H (où F et H sont des fonctions d'une seule variable et + une loi de composition interne, fonction de 2 variables) est étendue à l'expression F G H

(**FOURCHE**) où G est obligatoirement dyadique et F et H peuvent être, ensemble, soit monadiques, soit dyadiques.

- Le produit de composition est obtenu par la conjonction @
  $H=F \circ G$ (en mathématiques) se traduit par H =. F @ G (en J).
  Il est applicable aux verbes monadiques (fonctions d'une seule variable), mais aussi aux verbes dyadiques (fonctions de 2 variables).
- L'opération mathématique s'écrit en J : +/
- L'opération mathématique s'écrit en J : */
- Généralisation à tout verbe (ou pro-verbe) dyadique v : v/ (/ est un adverbe qui modifie le verbe v pour créer un nouveau verbe composite).
- Généralisation du "produit matriciel" (+/ .*) en (v1/ .v2) où v1 et v2 sont 2 verbes (ou pro-verbes) dyadiques.
- … Et beaucoup d'autres extensions….

COMPOSITIONS DE VERBES (ou pro-verbes)

Fourche rang de(u v w): _ _ _

```
  (u v w) y  ←→    (u y) v (w y)
x (u v w) y  ←→  (x u y) v (x w y)
```

Croc rang de(v w) : _ _ _

```
   (v w) y  ←→   y v w y
 x (v w) y  ←→   x v w y
```

Compose rang de(u&v): mv mv mv

```
    u&v  y  ←→        u v y
 x  u&v  y  ←→  (v x) u (v y)
```

Appose rang de(u&:v): _ _ _

```
   u&:v  y  ←→        u v y
 x u&:v  y  ←→  (v x) u (v y)
```

Atop rang de (u@v): mv lv rv

```
    u@v  y  ←→   u v y
 x  u@v  y  ←→ u x v y
```

At rang de (u@:v): _ _ _

```
   u@:v  y  ←→   u v y
 x u@:v  y  ←→ u x v y
```

Sous rang de(u&.v): mv mv mv

```
   u&.v  y  ←→  vi u v y      (vi = v^:_1)
 x u&.v  y  ←→  vi (v x) u (v y)
```

Sous rang de(u&.:v): _ _ _

```
    u&.:v    ←→  u&.(v"_)
```

Det – Prod mat rang de (u . v) : 2 _ _

Produit matriciel	`A (+/ . *) B`	NB. et produit scalaire
Déterminant	`(-/ . *) A`	
Permanent	`(+/ . *) A`	

Puissance fonctionnelle   rang de (u^:n) :   _ _ _

n entier

u^:n y ←→ u u …. u y   n fois si n entier positif ou nul

u^:_1 y ←→ inverse s'il existe (obverse)

u^:n y ←→ inverse appliqué |n fois si n entier négatif

n entier en boîte

u^:(<n)y ←→ u^:(i.n) y   si n entier positif ou nul

u^:(<n)y ←→ u^:_1^:(<|n) y   si n entier négatif

u^:(<m)y ←→ u^:(i.k) y   si m is _ ou ''

(k est le plus petit entier positif tel que (u^:(k-1) y) -: u^:k y)

x u^:n y ←→ x&u^:n y

x u^:(<n) y ←→ x&u^:(<n) y

n gérondif

```
  u^:(   v1`v2)y ←→ u^:(v1 y) (v2 y)
x u^:(v0`v1`v2)y ←→ (x v0 y)u^:(x v1 y) (x v2 y)
x u^:(   v1`v2)y ←→ x u^:([`v1`v2) y
```

Pro-conjonctions (divers produits de composition de 2 verbes ou pro-verbes)

Utilisation monadique

UVY =: 2 : 'u@v'	NB. **u(v y)**	**Rang mV**
VUY =: 2 : 'v@u'	NB. **v(u y)**	**Rang mU**
UY_VY =: 2 : '([:u])v]'	NB. **(u y)v y**	**Rang _**
VY_UY =: 2 : '([:v])u]'	NB. **(v y)u y**	**Rang _**
U_YVY =: 2 : 'u@(]v])'	NB. **u(y v y)**	**Rang _**
V_YUY =: 2 : 'v@(]u])'	NB. **v(y u y)**	**Rang _**
YU_VY =: 2 : ']u[:v]'	NB. **y u (v y)**	**Rang _**
YV_UY =: 2 : ']v[:u]'	NB. **y v (u y)**	**Rang _**

Utilisation dyadique

XUVY =: 2 : 'u v'	NB. **x u (v y)**	**Rang _ _**
YUVX =: 2 : '(u v)~'	NB. **y u (v x)**	**Rang _ _**
UY_VX =: 2 : 'v~u'	NB. **(u y) v x**	**Rang _ _**
UX_VY =: 2 : '(v~u)~'	NB. **(u x) v y**	**Rang _ _**
VY_UX =: 2 : 'u~v'	NB. **(v y) u x**	**Rang _ _**
VX_UY =: 2 : '(u~v)~'	NB. **(v x) u y**	**Rang _ _**
UXVY =: 2 : 'u@v'	NB. **u (x v y)**	**Rang _ _**
VXUY =: 2 : 'v@u'	NB. **v (x u y)**	**Rang _ _**
UYVX =: 2 : 'u@(v~)'	NB. **u (y v x)**	**Rang _ _**
VYUX =: 2 : 'v@(u~)'	NB. **v (y u x)**	**Rang _ _**

Résumé sur les compositions de verbes ou pro-verbes

Dans ce texte, **U, V, W** sont des verbes ou pro-verbes, x et y sont des noms ou pro-noms.
M_V, G_V , D_V sont les rangs Monadique, dyadique Gauche et dyadique Droit de **V**.

(U @: V) y	←→	**U (V y)**
x (U @: V) y	←→	**U (x V y)**
(U &: V) y	←→	**U (V y)**
x (U &: V) y	←→	**(V x) U (V y)**
(U @ V) y	←→	**(U @: V) ″ M_V y**
x (U @ V) y	←→	**x (U @: V) ″ G_V D_V y**
(U & V) y	←→	**(U @: V) ″ M_v y**
x (U & V) y	←→	**(V x) (U ″ (M_v , M_v)) y**
(U . V)	←→	**U @ (V ″ (1+G_V , _))**
(U . V)	←→	**(U @: V) ″ (1 + G_V , _)**
(+/ .*)	←→	**(+/ @: *) ″ 1 _**
x (U /.) y	←→	**(= x) (U @ #) y**
(U &. V) y	←→	**(V^:_1) U V y**
(U V) y	←→	**y U (V y)**
x (U V) y	←→	**x U (V y)**
(U V W) y	←→	**(U y) V (W y)**
x (U V W) y	←→	**(x U y) V (x W y)**
[y	←→	**y**
x [y	←→	**x**
] y	←→	**y**
x] y	←→	**y**
([: V W) y	←→	**V (W y)**
x ([: V W) y	←→	**V (x W y)**
([V W) y	←→	**y V (W y)**
x ([V W) y	←→	**x V (x W y)**
( ] V W) y	←→	**y V (W y)**
x ( ] V W) y	←→	**y V (x W y)**